Algorithm 463

Algorithms SCALE1, SCALE2, and SCALE3 for Determination of Scales on Computer Generated Plots [J6]

C.R. Lewart (Recd. 6 Aug. 1971 and 28 Jan. 1972)
Bell Telephone Laboratories, Incorporated, Holmdel, NJ 07733

Key Words and Phrases: plotting, scaling for plotting
CR Categories: 4.41
Language: Fortran

Description

Introduction. It is often desirable to plot computer generated output or obtain discrete distribution functions such as histograms automatically. In general the raw data does not lend itself directly to an easily readable presentation. The three related algorithms as presented here obtain readable linear or logarithmic scales with uniform interval sizes for users of various plot routines.

Readability. A readable linear scale is defined here as a scale with interval size a product of an integer power of 10 and 1, 2 or 5, and scale values integer multiples of the interval size.

A readable logarithmic scale on a display with uniform plotting intervals is defined here such that the ratio of adjoining scale values $DIST = 10^{(1/L+K)}$, where K and L are integers, with $1 \leq L \leq 10$; scale values are equal to $DIST^M$, where M is a set of successive integers.

The definition of readability used for *SCALE* 1 and *SCALE* 2 permits scale values such as:

−0.5, 0.0, 0.5, 1.0, . . .
1.24, 1.26, 1.28, . . .
100.0, 200.0, 300.0, . . . , etc.

It prohibits the following examples:

−1.0, 4.0, 9.0, . . .
1.2, 1.31, 1.42, . . .
0.0, 4.0, 8.0, 12.0, . . . , etc.

The definition of readability for logarithmic plots would permit scale values of 1, $\sqrt[3]{10}$, $(\sqrt[3]{10})^2$, 10, . . . , but disallow 1, $\sqrt{5}$, 5, $5\sqrt{5}$, 25, ...

Usage. A call of the form

*CALL SCALE*1 (*XMIN, XMAX, N, XMINP, XMAXP, DIST*)

where *XMIN* and *XMAX* are the minimum and maximum, respectively, of a given array and *N* a requested number of grid intervals will return a new minimum and maximum *XMINP* and *XMAXP* such that the range [*XMINP, XMAXP*] is the smallest range which will embrace the range [*XMIN, XMAX*] and simultaneously result in approximately *N* grid intervals, each of the length *DIST*. Interval *DIST* is selected by *SCALE*1 as the product of an integer power of 10 and 1, 2, or 5. *XMINP* and *XMAXP* are integer multiples of *DIST*.

In certain cases the number of plot intervals *N* has to be fixed. In particular, for plots generated by devices with relatively large pen increments, e.g. line printers or teletypewriters, *N* is restricted. For such cases *SCALE*2 for linear plots and *SCALE*3 for logarithmic plots have to be used.

*SCALE*2 with the same arguments as *SCALE*1 differs from *SCALE*1 in that *XMINP* and *XMAXP* are determined such that exactly *N* grid intervals will result; as a consequence the range [*XMINP, XMAXP*] will in general be less economical than that obtained by *SCALE*1. Parameters *DIST, XMINP,* and *XMAXP* will still satisfy requirements specified for *SCALE*1, namely *DIST* will be an integer power of 10 times 1, 2, or 5; and *XMINP* and *XMAXP* will be integer multiples of *DIST*.

*SCALE*3 with the same arguments as *SCALE*1 will set *XMINP* and *XMAXP* such that *N* logarithmic uniformly spaced grid intervals will cover the range [*XMIN, XMAX*]. *DIST* will be the ratio of adjacent grid line values.

*SCALE*3 selects *DIST* as $10^{(1/L+K)}$, where K and L are integers and $1 \leq L \leq 10$. *XMINP* and *XMAXP* are selected so that $XMINP = DIST^j$, and $XMAXP = DIST^l$ where j and l are integers.

Calling *SCALE*1, *SCALE*2, or *SCALE*3 will approximately center the range [*XMIN, XMAX*] between *XMINP* and *XMAXP*. *SCALE*1, having determined *DIST*, selects the most economical limits, i.e. $(XMIN - DIST) < XMINP \leq XMIN$ and $XMAX \leq XMAXP < (XMAX + DIST)$. *SCALE*2 and *SCALE*3 select limits to minimize $(XMAXP - XMAX)$ and $(XMIN - XMINP)$ without necessarily satisfying the previous inequalities, but subject to the constraints of a fixed number of intervals.

The actual number of intervals N_a, determined from the outputs returned by *SCALE*1 is as follows:

$$N_a = (XMAXP - XMINP)/DIST.$$

N_a may be slightly larger or smaller than N as shown by the following inequality:

$$(N/\sqrt{2.5}) < N_a < (N \times \sqrt{2.5} + 2).$$

N_a will always equal N if *SCALE*2 or *SCALE*3 is called.

Round-off considerations. The three algorithms compensate for the computer round-off to assure that *XMIN* and *XMAX* are within the range [*XMINP, XMAXP*]. A normalized parameter *DEL* is introduced to serve as a narrow gate around the minimum *XMIN* and the maximum *XMAX* to avoid an unnecessarily large range [*XMINP, XMAXP*] caused by computer round-off. For example, if $DEL = 0.0001$, $N = 3$ and *SCALE*1 or *SCALE*2 is called, *XMINP* of 1.0 and *XMAXP* of 4.0 will result for $0.9999 < XMIN \leq 1.0001$ and $3.9999 \leq XMAX < 4.0001$. *DEL* is normalized to the interval size and should satisfy the following inequality:

$$A < DEL < (B \times N)/C,$$

where A is the round-off expected from a division and float operation, B is the minimum increment of the plotting device in inches, N is the number of intervals on the plot, and C is the plot size in inches. For example, using single precision *REAL**4 variables (IBM 360): $A \sim 0.0000002$; for a precision flat bed plotter: $B = 0.002$, $C = 50.0$. Assuming $N = 10$ the following inequality is obtained:

$$0.0000002 < DEL < 0.0004.$$

It is obvious from this inequality that in practical cases the range of permissible values of *DEL* is so large that *DEL* is quite insensitive to the type of plotter and the type of computer used.

Examples

*SCALE*1

XMIN	*XMAX*	*N*	*XMINP*	*XMAXP*	*DIST*	*Actual No. of Intervals*
−3.1	11.1	5	−4.0	12.0	2.0	8
5.2	10.1	5	5.0	11.0	1.0	6
−12000	−100	9	−12000	0	1000	12

*SCALE*2

XMIN	*XMAX*	*N*	*XMINP*	*XMAXP*	*DIST*	*Actual No. of Intervals*
−3.1	11.1	5	−5.0	20.0	5.0	5
5.2	10.1	5	4.0	14.0	2.0	5
−12000	−100	9	−14000	4000	2000	9

SCALE 3

XMIN	*XMAX*	*N*	*XMINP*	*XMAXP*	*DIST*	*Actual No. of Intervals*
1.8	125.0	10	1.58	158.49	1.58 ($=\sqrt[5]{10}$)	10
0.1	10.0	2	0.1	10.0	10.0	2
0.1	1500.0	4	0.077	2154.4	12.92 ($=10^{(1+1/9)}$)	4

Algorithm

```
      SUBROUTINE SCALE1(XMIN, XMAX, N, XMINP, XMAXP, DIST)
C ANSI FØRTRAN
C GIVEN XMIN,XMAX AND N SCALE1 FINDS A NEW RANGE XMINP AND
C XMAXP DIVISIBLE INTØ APPRØXIMATELY N LINEAR INTERVALS
C ØF SIZE DIST
C VINT IS AN ARRAY ØF ACCEPTABLE VALUES FØR DIST (TIMES
C AN INTEGER PØWER ØF 10)
C SQR IS AN ARRAY ØF GEØMETRIC MEANS ØF ADJACENT VALUES
C ØF VINT, IT IS USED AS BREAK PØINTS TØ DETERMINE
C WHICH VINT VALUE TØ ASSIGN TØ DIST
      DIMENSIØN VINT(4), SQR(3)
      DATA VINT(1), VINT(2), VINT(3), VINT(4)/1., 2., 5., 10./
      DATA SQR(1), SQR(2), SQR(3)/1.414214, 3.162278, 7.071068/
C CHECK WHETHER PRØPER INPUT VALUES WERE SUPPLIED
      IF (XMIN.LT.XMAX .AND. N.GT.0) GØ TØ 10
      WRITE (6,99999)
99999 FØRMAT(34H IMPRØPER INPUT SUPPLIED TØ SCALE1)
      RETURN
C DEL ACCØUNTS FØR CØMPUTER RØUND-ØFF
C DEL SHØULD BE GREATER THAN THE RØUND-ØFF EXPECTED FRØM
C A DIVISIØN AND FLØAT ØPERATIØN, IT SHØULD BE LESS THAN
C THE MINIMUM INCREMENT ØF THE PLØTTING DEVICE USED BY
C THE MAIN PRØGRAM (IN.) DIVIDED BY THE PLØT SIZE (IN.)
C TIMES NUMBER ØF INTERVALS N
   10 DEL = .00002
      FN = N
C FIND APPRØXIMATE INTERVAL SIZE A
      A = (XMAX-XMIN)/FN
      AL = ALØG10(A)
      NAL = AL
      IF (A.LT.1.) NAL = NAL - 1
C A IS SCALED INTØ VARIABLE NAMED B BETWEEN 1 AND 10
      B = A/10.**NAL
C THE CLØSEST PERMISSIBLE VALUE FØR B IS FØUND
      DØ 20 I=1,3
        IF (B.LT.SQR(I)) GØ TØ 30
   20 CONTINUE
      I = 4
C THE INTERVAL SIZE IS CØMPUTED
   30 DIST = VINT(I)*10.**NAL
      FM1 = XMIN/DIST
      M1 = FM1
      IF (FM1.LT.0.) M1 = M1 - 1
      IF (ABS(FLØAT(M1)+1.-FM1).LT.DEL) M1 = M1 + 1
C THE NEW MINIMUM AND MAXIMUM LIMITS ARE FØUND
      XMINP = DIST*FLØAT(M1)
      FM2 = XMAX/DIST
      M2 = FM2 + 1.
      IF (FM2.LT.(-1.)) M2 = M2 - 1
      IF (ABS(FM2+1.-FLØAT(M2)).LT.DEL) M2 = M2 - 1
      XMAXP = DIST*FLØAT(M2)
C ADJUST LIMITS TØ ACCØUNT FØR RØUND-ØFF IF NECESSARY
      IF (XMINP.GT.XMIN) XMINP = XMIN
      IF (XMAXP.LT.XMAX) XMAXP = XMAX
      RETURN
      END

      SUBRØUTINE SCALE2(XMIN, XMAX, N, XMINP, XMAXP, DIST)
C ANSI FØRTRAN
C GIVEN XMIN,XMAX AND N SCALE2 FINDS A NEW RANGE XMINP AND
C XMAXP DIVISIBLE INTØ EXACTLY N LINEAR INTERVALS ØF SIZE
C DIST, WHERE N IS GREATER THAN 1
      DIMENSIØN VINT(5)
      DATA VINT(1), VINT(2), VINT(3), VINT(4), VINT(5)/1., 2.,
     * 5., 10., 20./
C CHECK WHETHER PRØPER INPUT VALUES WERE SUPPLIED
      IF (XMIN.LT.XMAX .AND. N.GT.1) GØ TØ 10
      WRITE (6,99999)
99999 FØRMAT(34H IMPRØPER INPUT SUPPLIED TØ SCALE2)
      RETURN
   10 DEL = .00002
      FN = N
C FIND APPRØXIMATE INTERVAL SIZE A
      A = (XMAX-XMIN)/FN
      AL = ALØG10(A)
      NAL = AL
      IF (A.LT.1.) NAL = NAL - 1
C A IS SCALED INTØ VARIABLE NAMED B BETWEEN 1 AND 10
      B = A/10.**NAL
C THE CLØSEST PERMISSIBLE VALUE FØR B IS FØUND
      DØ 20 I=1,3
        IF (B.LT.(VINT(I)+DEL)) GØ TØ 30
   20 CØNTINUE
      I = 4
C THE INTERVAL SIZE IS CØMPUTED
   30 DIST = VINT(I)*10.**NAL
      FM1 = XMIN/DIST
      M1 = FM1
      IF (FM1.LT.0.) M1 = M1 - 1
      IF (ABS(FLØAT(M1)+1.-FM1).LT.DEL) M1 = M1 + 1
C THE NEW MINIMUM AND MAXIMUM LIMITS ARE FØUND
      XMINP = DIST*FLØAT(M1)
      FM2 = XMAX/DIST
      M2 = FM2 + 1.
      IF (FM2.LT.(-1.)) M2 = M2 - 1
      IF (ABS(FM2+1.-FLØAT(M2)).LT.DEL) M2 = M2 - 1
      XMAXP = DIST*FLØAT(M2)
C CHECK WHETHER A SECØND PASS IS REQUIRED
      NP = M2 - M1
      IF (NP.LE.N) GØ TØ 40
      I = I + 1
      GØ TØ 30
   40 NX = (N-NP)/2
      XMINP = XMINP - FLØAT(NX)*DIST
      XMAXP = XMINP + FLØAT(N)*DIST
C ADJUST LIMITS TØ ACCØUNT FØR RØUND-ØFF IF NECESSARY
      IF (XMINP.GT.XMIN) XMINP = XMIN
      IF (XMAXP.LT.XMAX) XMAXP = XMAX
      RETURN
      END

      SUBRØUTINE SCALE3(XMIN, XMAX, N, XMINP, XMAXP, DIST)
C ANSI FØRTRAN
C GIVEN XMIN,XMAX AND N, WHERE N IS GREATER THAN 1, SCALE3
C FINDS A NEW RANGE XMINP AND XMAXP DIVISIBLE INTØ EXACTLY
C N LØGARITHMIC INTERVALS, WHERE THE RATIØ ØF ADJACENT
C UNIFØRMLY SPACED SCALE VALUES IS DIST
      DIMENSIØN VINT(11)
      DATA VINT(1), VINT(2), VINT(3), VINT(4), VINT(5), VINT(6),
     * VINT(7), VINT(8), VINT(9), VINT(10), VINT(11)/10., 9.,
     * 8., 7., 6., 5., 4., 3., 2., 1., .5/
C CHECK WHETHER PRØPER INPUT VALUES WERE SUPPLIED
      IF (XMIN.LT.XMAX .AND. N.GT.1 .AND. XMIN.GT.0.) GØ TØ 10
      WRITE (6,99999)
99999 FØRMAT(34H IMPRØPER INPUT SUPPLIED TØ SCALE3)
      RETURN
   10 DEL = .00002
C VALUES ARE TRANSLATED FRØM THE LINEAR INTØ LØGARITHMIC
C REGIØN
      XMINL = ALØG10(XMIN)
      XMAXL = ALØG10(XMAX)
      FN = N
C FIND APPRØXIMATE INTERVAL SIZE A
      A = (XMAXL-XMINL)/FN
      AL = ALØG10(A)
      NAL = AL
      IF (A.LT.1.) NAL = NAL - 1
C A IS SCALED INTØ VARIABLE NAMED B BETWEEN 1 AND 10
      B = A/10.**NAL
C THE CLØSEST PERMISSIBLE VALUE FØR B IS FØUND
      DØ 20 I=1,9
        IF (B.LT.(10./VINT(I)+DEL)) GØ TØ 30
   20 CØNTINUE
      I = 10
C THE INTERVAL SIZE IS CØMPUTED
   30 DISTL = 10.**(NAL+1)/VINT(I)
      FM1 = XMINL/DISTL
      M1 = FM1
      IF (FM1.LT.0.) M1 = M1 - 1
      IF (ABS(FLØAT(M1)+1.-FM1).LT.DEL) M1 = M1 + 1
C THE NEW MINIMUM AND MAXIMUM LIMITS ARE FØUND
      XMINP = DISTL*FLØAT(M1)
      FM2 = XMAXL/DISTL
      M2 = FM2 + 1.
      IF (FM2.LT.(-1.)) M2 = M2 - 1
      IF (ABS(FM2+1.-FLØAT(M2)).LT.DEL) M2 = M2 - 1
      XMAXP = DISTL*FLØAT(M2)
      NP = M2 - M1
C CHECK WHETHER ANØTHER PASS IS NECESSARY
      IF (NP.LE.N) GØ TØ 40
      I = I + 1
      GØ TØ 30
   40 NX = (N-NP)/2
      XMINP = XMINP - FLØAT(NX)*DISTL
      XMAXP = XMINP + FLØAT(N)*DISTL
C VALUES ARE TRANSLATED FRØM THE LØGARITHMIC INTØ THE LINEAR
C REGIØN
      DIST = 10.**DISTL
      XMINP = 10.**XMINP
      XMAXP = 10.**XMAXP
C ADJUST LIMITS TØ ACCØUNT FØR RØUND-ØFF IF NECESSARY
      IF (XMINP.GT.XMIN) XMINP = XMIN
      IF (XMAXP.LT.XMAX) XMAXP = XMAX
      RETURN
      END
```